

# Some criteria for implementations of conjunction and disjunction in microKanren

Anonymous Author(s)

## Abstract

We describe how an additional language feature—namely variadic, “rest argument” functions—can lead to a more powerful kernel language. We implement conjunction and disjunction functions that preserve search order and that do not create superfluous closures. The result is a shorter and simpler overall implementation that is more elegant.

## ACM Reference Format:

Anonymous Author(s). 2022. Some criteria for implementations of conjunction and disjunction in microKanren. In *Proceedings of miniKanren Workshop (mKW '22)*. ACM, New York, NY, USA, 6 pages. <https://doi.org/XXXXXXX.XXXXXXX>

## 1 Introduction

microKanren [4] is a compact approach to implementing a relational programming language. The microKanren approach has worked out well as a tool for understanding the guts of a relational programming language through studying its implementation. The microKanren reimplementations separates surface syntax macros from function definitions. In doing so, the authors hoped this separation would simultaneously aid future would-be implementers when studying the source code, and also that the functional core would make the language easier to port to other functional hosts. To support both those efforts, they also chose to program in a deliberately small and workaday set of Scheme primitives.

This note shows how an implementer using a host language with variadic (any number of arguments) functions can build a somewhat more powerful kernel language and obviate macros for conjunction and disjunction. Until now there was a large gap between those microKanren implementations in languages with expressive macro systems and those without; variadic functions permit an intermediate point in the language design space. Even implementers in

languages *with* macro systems might choose to replace existing less powerful language primitives by our comparatively short but more powerful variants. Beyond that, this more powerful kernel language obviates several macros in the desugaring layer, and eventually leads us to see merit in modifying the core language’s syntax. This is, of course, subjective.

Beyond being a conceptually simpler foundation for a full Kanren language at the level of *The Reasoned Schemer, Second Edition*, this implementation appears to be more *efficient*. Right-associative conjunction is already conjectured in miniKanren implementation folklore to be an inefficient design choice. We were pleasantly surprised that thinking about the expressivity of primitives and code improvement “horse sense” led us to the same improved design.

In Section 2, we briefly revisit microKanren implementations and illustrate why surface syntax macros had seemed practically mandatory. In Section 3, we implement conjunction and disjunction, and in Section 4 we discuss the reimplementation of the impure operators. We close with a discussion of some performance impacts of these implementation choices, and consider how Kanren language implementers outside of the Scheme family might benefit from these alternatives.

## 2 All Aboard!

We assume the reader is familiar with the miniKanren language as described in *The Reasoned Schemer, Second Edition* and in particular with microKanren implementations. See [minikanren.org](http://minikanren.org) for many implementations across multiple host languages.

The world’s shortest subway system is Haifa’s Carmelit, with only six stations. The system is a line, so its trains travel back and forth. If we wanted to describe the order in which we rode the train from the beginning to the end, we could use the `carmelit-subway` relation in Listing 1. We express this relation using the more compact miniKanren syntax because unfolding this relation into binary conjunctions and disjunctions would be painful. We certainly *could* write it out by hand—in fact, we could write it out many ways. We could nest those conjunctions to the left, or to the right, or try and reduce the indentation by trying to keep them somewhat balanced: the program itself does not seem to obviously encourage one particular choice.

---

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from [permissions@acm.org](mailto:permissions@acm.org).

mKW '22, September 15, 2022, Ljubljana, Slovenia

© 2022 Association for Computing Machinery.

ACM ISBN 978-1-4503-XXXX-X/18/06...\$15.00

<https://doi.org/XXXXXXX.XXXXXXX>

```

111 (defrel (carmelit-subway a b c d e f)
112   (conde
113     ((= a 'carmel-center)
114      (= b 'golomb)
115      (= c 'masada)
116      (= d 'haneviim)
117      (= e 'hadar-city-hall)
118      (= f 'downtown))
119     ((= a 'downtown)
120      (= b 'hadar-city-hall)
121      (= c 'haneviim)
122      (= d 'masada)
123      (= e 'golomb)
124      (= f 'carmel-center))))

```

**Listing 1.** A miniKanren implementation of the Carmelit subway.

```

129 (define-syntax conde
130   (syntax-rules ()
131     ((conde (g ...) ...)
132      (disj (conj g ...) ...))))
133
134 (define-syntax conj
135   (syntax-rules ()
136     ((conj) #s)
137     ((conj g)
138      ((conj g0 g ...) (conj2 g0 (conj g ...))))))
139
140 (define ((conj2 g1 g2) s)
141   (append-map∞ g2 (g1 s)))
142
143 (define-syntax disj
144   (syntax-rules ()
145     ((disj) #u)
146     ((disj g)
147      ((disj g0 g ...) (disj2 g0 (disj g ...))))))
148
149 (define ((disj2 g1 g2) s)
150   (append∞ (g1 s) (g2 s)))
151

```

**Listing 2.** Macro based implementations of conj and disj.

Listing 2 shows typical macro-based implementations of conde and the underlying conjunction and disjunction operations. Here, #s and #u represent primitive goals that unconditionally succeed and fail, respectively.

### 3 conj and disj logical goal constructors

The conj and disj macros of Listing 2 are not *quite* defined as simple recursions over their binary functional primitives. The definition of zero-way conjunction (disjunction) is independent of the unfolding of conj<sub>2</sub> (disj<sub>2</sub>). In a sense

```

166 (define ((disj g . gs) s)
167   (D (g s) gs s))
168
169 (define (D s∞ gs s)
170   (cond
171     ((null? gs) s∞)
172     (else
173      (append∞
174       (D ((car gs) s) (cdr gs) s))))))
175

```

**Listing 3.** Re-implementation of disj.

```

176
177
178 (define ((conj g . gs) s)
179   (C gs (g s)))
180
181 (define (C gs s∞)
182   (cond
183     ((null? gs) s∞)
184     (else
185      (C (cdr gs)
186       (append-map∞ (car gs) s∞))))))
187

```

**Listing 4.** Re-implementation of conj.

the conj and disj macros confuse and entangle primitive success and failure goals with those recursive unfoldings in terms of the binary operators.

Those zero-way logical operation base cases don't add much. The programmer who tries to write an elegant, efficient solution to a pure relational programming task would not encounter these additional base cases in conde expressions. The programmer knows statically how such goals should behave, so there is no benefit to executing them. A conjunction of no goals would simply succeed, and a disjunction of no goals would simply fail. Truthfully, only the impure conda operator seems to require these additional base cases (discussed further in Section 4). They are for all intents and purposes superfluous.

Here, we want to implement conjunction and disjunction over one-or-more goals as functions. These implementations will rely on variadic function syntax, but should not require apply or rely on the binary conj<sub>2</sub> and disj<sub>2</sub>. Further, they should not build any extraneous closures: unnecessarily building closures at runtime is always a bad idea.

Listings 3 and 4 show our new implementations. Mandating one-or-more arguments lets us re-implement these operators as shallow wrappers over simple folds. The first step in each is merely to remove the rest argument gs and act as if there were no need for a rest argument. That is, all of our focus will be on the *list*, gs. Unlike D, the function C does not take in the state s; the help procedure does not need the state for conjunction. In each recursive call, we accumulate by mapping (using the special delaying implementation of

append-map for miniKanren streams) the next goal in the list. This left-fold implementation of conjunction therefore left-associates the conjuncts.

### 3.1 Implementation correctness

Equipped with variadic functions, a developer might discover these definitions as follows. We start from the definition of the disj macro in Listing 7. At the cost of an apply, we can build the corresponding explicitly recursive disj function.

```
(define-syntax disj
  (syntax-rules ()
    ((disj g) g)
    ((disj g0 g1 g ...)
     (disj2 g0 (disj g1 g ...))))

(define (disj g . gs)
  (cond
    ((null? gs) g)
    (else (disj2 g (apply disj gs)))))
```

**Listing 5.** Derivation of disj function definition from macro.

Since disj produces and consumes goals, we can  $\eta$  expand the definition by a curried parameter *s*. We then split disj into two mutually-recursive procedures, to build the variant in Listing 6.

```
(define ((disj g . gs) s)
  (D g gs s))

(define (D g gs s)
  (cond
    ((null? gs) (g s))
    (else ((disj2 g (apply disj gs)) s))))
```

**Listing 6.** An  $\eta$  expanded and split definition of disj.

We can replace the call to disj<sub>2</sub> in by its definition in terms of append<sup>∞</sup> and perform a trivial  $\beta$ -reduction. The explicit *s* argument suggests removing the call to apply and make *D* recur into itself. The result is the version of *D* in Listing 7.

```
(define (D g gs s)
  (cond
    ((null? gs) (g s))
    (else
     (append∞ (g s)
                (D (car gs) (cdr gs) s)))))
```

**Listing 7.** Derivation of disj function definition.

```
(define-syntax conj
  (syntax-rules ()
    ((conj g) g)
    ((conj g g1 gs ...)
     (conj (conj2 g g1) gs ...))))

(define (conj g . gs)
  (cond
    ((null? gs) g)
    (else
     (apply conj
              (cons (conj2 g (car gs)) (cdr gs))))))
```

**Listing 8.** conj<sub>2</sub>-based conj function and macro.

We combine *g* and *s* in each clause; this suggests constructing that stream in disj and passing it along. This final change results in the definition in Listing 3.

We can derive the definition of conj from Listing 4 via a similar process. Starting with the variadic function definition of Listing 8, we first  $\eta$ -expand and split the definition.

```
(define ((conj g . gs) s)
  (C g gs s))

(define (C g gs s)
  (cond
    ((null? gs) (g s))
    (else
     ((apply conj
              (cons (conj2 g (car gs)) (cdr gs))
                    s))))))
```

We will substitute the definitions of conj and conj<sub>2</sub>.

```
(define ((conj g . gs) s)
  (C g gs s))

(define (C g gs s)
  (cond
    ((null? gs) (g s))
    (else
     (C (λ (s) (append-map∞ (car gs) (g s)))
         (cdr gs)
         s))))
```

Finally, since *C* only needs *s* to *build* the stream, we can assemble the stream on the way in—instead of passing in *g* and *s* separately, we pass in their combination as a stream. The function is tail recursive, we can change the signature in the one and only external call and the recursive call. We show the result in Listing 4.

Both the variadic function based and the macro based versions of Listing 8 use a left fold over the goals, whereas the versions of disj use a right fold. This is not an accident. It is miniKanren folklore that left associating conjunctions tends

to improve miniKanren performance. A full, thorough, explanation of what we know about this behavior is outside the scope of our work here, but see discussions by Rosenblatt [nowhere] and Ballantyne [nowhere]. The matter seems so significant that the community might consider reclassifying left-associative conjunction as a matter of correctness rather than an optimization, as in “tail call optimization” vs. “Properly Implemented Tail Call Handling” [2]. However, we have generally resorted to small step visualizations of the search tree to demonstrate why that might be. The authors believe it is *worth further consideration* if we can make an equally compelling argument for this preference through equational reasoning. Listing 9 shows a right-fold variant of conj. The choice to fold left becomes a little more obvious after we  $\eta$ -expand, unfold to a recursive help function, substitute in the definition of conj<sub>2</sub>, and  $\beta$ -reduce.

```
(define (conj g . gs)
  (cond
    ((null? gs) g)
    (else (conj2 g (apply conj gs)))))
```

**Listing 9.** A right-fold variant of conj.

```
(define ((conj g . gs) s)
  (C gs (g s)))

(define (C g gs s)
  (cond
    ((null? gs) (g s))
    (else (append-map $\infty$  (apply C gs) (g s)))))
```

Here, we cannot (easily) replace the apply call by a recursive call to C, because we are still waiting for an s. We can only abstract over s and wait.

```
(define (C gs s $\infty$ )
  (cond
    ((null? gs) s $\infty$ )
    (else
     (append-map $\infty$ 
      ( $\lambda$  (s) (C (cdr gs) ((car gs) s)))
      s $\infty$ ))))
```

Since we know that any call to append-map $\infty$  we construct will always yield a result, the version in Listing 4 is tail recursive. The equivalent right-fold implementation needs to construct a closure for every recursive call. If we want to implement a variadic version that does not rely on a primitive conj<sub>2</sub> and does not resort to apply, we have the two aforementioned choices. Basic programming horse sense suggests the variant from Listing 4.

Though this note mainly concerns the choice to implement surface language behavior as functions, it may also point to these as more natural user-level primitives than

```
(defrel (carmelit-subway a b c d e f)
  (disj
    (conj (== a 'carmel-center)
          (== b 'golomb)
          (== c 'masada)
          (== d 'haneviim)
          (== e 'hadar-city-hall)
          (== f 'downtown))
    (conj (== a 'downtown)
          (== b 'hadar-city-hall)
          (== c 'haneviim)
          (== d 'masada)
          (== e 'golomb)
          (== f 'carmel-center)))))
```

**Listing 10.** A reimplemented Carmelit subway without conde.

conde. An implementation could choose to forego conde and provide just those underlying logical primitives conj and disj to the user.

## 4 Cleaning up the Impure Operators

Some programmers would be perfectly satisfied just using ifte directly. But just as the standard forked if begat McCarthy’s if notation and cond, we can suspect that a programmer would eventually feel the need for a nested implementation.

The operators conda and conde look superficially similar, syntactically. Semantically though, the disjunction and nested “if-then-else” behaviors are quite different, and implementing the desired behavior for this conda from existing pieces raises some oddities. The definition of conda (see Listing 11) rules out zero-way conjunctions of goals in the clause body as well as zero-way disjunctions of such clauses. A conda that permits one-or-more-goals in each clause is the one and only place in the whole language implementation that may require a nullary conjunction of goals. So this soft-cut operator seems to force both nullary conjunction and those primitive goals #s and #u into the language.

There are alternative choices one could consider.

1. Syntactically mandate that all clauses in conda contain at least two goals.
2. Syntactically mandate that all clauses *except the final default clause* contain at least two goals.
3. Introduce a special clause of the conda macro specifically for “if then” clauses with a single goal.
4. Unconditionally add an #s goal to each clause during macro expansion.

Each of these choices can, implicitly or explicitly, force additional unneeded executions of unwanted goals. Here too, removing macros and relying on variable arity function definition provides a more elegant solution.

```

441 (define-syntax conda
442   (syntax-rules ()
443     ((conda (g0 g ...) (conj g0 g ...))
444      ((conda (g0 g ...) ln ...)
445       (ifte g0 (conj g ...) (conda ln ...)))))
446
447 (define ((ifte g1 g2 g3) s)
448   (let loop ((s∞ (g1 s)))
449     (cond
450      ((null? s∞) (g3 s))
451      ((pair? s∞)
452       (append-map g2 s∞))
453      (else (lambda ()
454               (loop (s∞)))))))

```

**Listing 11.** A typical implementation of conda.

```

458 (define ((conda q a . q-and-a*) s)
459   (A (q s) a q-and-a* s))
460
461 (define (A s∞ a q-and-a* s)
462   (cond
463    ((null? s∞)
464     (cond
465      ((null? (cdr q-and-a*)) ((car q-and-a*) s))
466      (else (A ((car q-and-a*) s)
467                (cadr q-and-a*)
468                (cddr q-and-a*) s))))))
469   ((pair? s∞) (append-map∞ a s∞))
470   (else (lambda () (A (s∞) a q-and-a* s)))))

```

**Listing 12.** A functional conda implementation.

The implementation in Listing 12 includes the delay-and-restart behavior of ifte together with conda’s logical cascade.

#### 4.1 condu

We have not yet addressed condu. We could implement this for functional programmers with a variant of `??`. Instead, we choose to make once a primitive higher-order goal. This relies on a local `let-bound` named recursive procedure.

## 5 Conclusion

narrows the gap between implementations in functional host languages with and without macro systems.

More languages than ever before support variable arity functions/methods (aka `varargs` aka `slurpy` methods), including Raku, Java, and Ruby, to name just a few. Many of these languages also support polyvariadic functions. The authors hope that this document helps implementers build more elegant, expressive and efficient Kanrens in their chosen host languages.

```

496 (define (once g)
497   (lambda (s)
498     (let loop ((s∞ (g s)))
499       (cond
500        ((null? s∞) '())
501        ((pair? s∞)
502         (cons (car s∞) '()))
503        (else (lambda ()
504                  (loop (s∞)))))))

```

**Listing 13.** A functional once implementation.

Languages with variadic but without polyvariadic (that is to say “at least  $k$  arguments”) functions—how do they fare here?

This is closer to the design of Prolog, where the user represents conjunction of goals in the body of a clause with a comma and disjunction, either implicitly in listing various clauses or explicitly with a semicolon.

Languages without expressive macro systems would still have to resort to exposing implementations of streams and *some* lower level operations like introducing variables one at a time. But!

We take it as granted that conjunction and disjunction themselves are sufficiently high-level operators for a surface language and that the 0-element base cases are at best unnecessary and likely undesirable.

Given the opportunity to define a surface language and its desugaring, we really shouldn’t tempt the programmer by making undesirable programs representable when we can avoid doing so.

Of course, no implementer *needs* a macro system to implement a shallow embedding of an LP language. In our implementations, we still use macros to implement `defrel` and `fresh`, the former to prevent exposing the implementation of streams and delays, and the latter both for the added efficiency and to avoid the awkwardness of introducing each variable one at a time.

From time to time we find that the usual miniKanren implementation is itself lower-level than we would like to program with relations. Expert miniKanren programmers use macros to extend the language yet again, as with `matche`, and also with wholly more expressive and highly optimized language forms, as in the original Kanren. It is nice to find we can spread some of that extra expressiveness to the wider Kanren language community.

Even implementers using and programmers studying implementations written in languages *with* sufficiently expressive macro systems can benefit from these improvements. There are benefits to using a limited set of features in a language. Based on an evaluation of `minikanren.org`, earlier microKanren implementers measured their results by the



number of lines in the core implementation, the number of additional lines for the surface syntax, and the limited features they required from the host. So replacing

Early microKanren implementations restrict themselves to syntax-rules macros. Several more powerful syntax constructions over miniKanren may rely on more expressive macro systems, including Ballantyne's `minikanren-ee` [1].

This note shows that at the cost of one additional feature—namely variadic functions—implementers provide a somewhat more powerful core language and significantly reduce the need for macros in implementing a language as expressive as the full microKanren + Macros approach provides.

Our desugaring macros would have worked whether we left or right associated. Removing a degree of freedom in the implementation, and forcing ourselves to program the solution functionally, led us to what seems like the right solution.

## References

- [1] Michael Ballantyne, Alexis King, and Matthias Felleisen. 2020. Macros for domain-specific languages. *Proceedings of the ACM on Programming Languages*, 4, OOPSLA, 1–29.
- [2] Matthias Felleisen. [n. d.] Re: question about tail recursion. (). <https://lists.racket-lang.org/users/archive/2014-August/063844.html>.
- [3] Daniel P. Friedman, William E. Byrd, Oleg Kiselyov, and Jason Hemann. 2018. *The Reasoned Schemer, Second Edition*. The MIT Press, (Mar. 2018). ISBN: 0-262-53551-3. [mitpress.mit.edu/books/reasoned-schemer-0](http://mitpress.mit.edu/books/reasoned-schemer-0).
- [4] Jason Hemann and Daniel P. Friedman. 2013. Mkanren: a minimal functional core for relational programming. In *Scheme 13*. <http://scheme-workshop.org/2013/papers/HemannMuKanren2013.pdf>.